Appendix: Source Code

The comments from the illustration (Figure 7) are also in the code below.

This code can be executed in SE38, provided the two InfoObjects ZPRFT_CTR and ZMUNIT are defined as above in the system.

However, in a real, live BW system it makes more sense to use this code in extractors or staging routines.


TABLES:
     /bic/qzmunit,
     /bic/qzprft_ctr.

TYPES:
    BEGIN OF t_zprft_ctr_unit_key,
         zprft_ctr TYPE /bic/qzprft_ctr-/bic/zprft_ctr,
         zmunit    TYPE /bic/qzmunit-/bic/zmunit,
    END OF t_zprft_ctr_unit_key.

DATA:
*ZPRFT_CTR
    lt_zprft_ctr TYPE Sorted TABLE OF /bic/qzprft_ctr 

with unique key /bic/zprft_ctr  datefrom dateto /bic/zmunit, " this declaration ensures performance in the "loop at where" statement below
    lt_zprft_ctr_tmp TYPE TABLE OF /bic/qzprft_ctr,
    lt_zprft_ctr_unit_key TYPE TABLE OF t_zprft_ctr_unit_key,
    ls_zprft_ctr_tmp TYPE /bic/qzprft_ctr,
*ZMUNIT
    lt_zmunit TYPE sorted TABLE OF /bic/qzmunit with unique key /bic/zmunit datefrom dateto, " this declaration ensures performance in the "loop at where" statement below
    lt_zmunit_tmp TYPE TABLE OF /bic/qzmunit,
    ls_zmunit_tmp TYPE /bic/qzmunit,
*ZTARGET
    lt_ztarget TYPE TABLE OF /bic/qztarget,
    ls_ztarget TYPE /bic/qztarget,
*profit center unit key
    ls_zprft_ctr_unit_key TYPE t_zprft_ctr_unit_key,

*FLAGS FOR PROVIDE STATEMENT
l_flg1 TYPE rs_bool,
l_flg2 TYPE rs_bool.


FIELD-SYMBOLS:
    <ls_zprft_ctr> TYPE /bic/qzprft_ctr,
    <ls_zprft_ctr_tmp> TYPE /bic/qzprft_ctr,
    <ls_zprft_ctr_unit_key> TYPE t_zprft_ctr_unit_key,
    <ls_zmunit> TYPE /bic/qzmunit,
    <ls_ztarget> TYPE /bic/qztarget.

***********************
* 1. Select data from InfoObjects into internal tablesselect data from infoobjects into internal tables
SELECT * FROM /bic/qzprft_ctr INTO CORRESPONDING FIELDS OF TABLE lt_zprft_ctr.
SELECT * FROM /bic/qzmunit INTO CORRESPONDING FIELDS OF TABLE lt_zmunit.


* 2. 2.
Loop over all profit centers and select those that have a management unit assigned. These are run through the PROVIDE statement.

* All other profit center records are attached directly to the target table.
IF lt_zprft_ctr IS NOT INITIAL.
  LOOP AT lt_zprft_ctr ASSIGNING <ls_zprft_ctr>.
    IF <ls_zprft_ctr>-/bic/zprft_ctr IS NOT INITIAL AND <ls_zprft_ctr>-/bic/zmunit IS NOT INITIAL.
      ls_zprft_ctr_unit_key-zprft_ctr = <ls_zprft_ctr>-/bic/zprft_ctr.
      ls_zprft_ctr_unit_key-zmunit = <ls_zprft_ctr>-/bic/zmunit.
      APPEND ls_zprft_ctr_unit_key TO lt_zprft_ctr_unit_key.
    ELSE. "no provide required => direct append to target table
      ls_ztarget-/bic/ztarget = <ls_zprft_ctr>-/bic/zprft_ctr.
      MOVE-CORRESPONDING <ls_zprft_ctr> TO ls_ztarget.
      APPEND ls_ztarget TO lt_ztarget.
    ENDIF.
  ENDLOOP.
ENDIF.
SORT lt_zprft_ctr_unit_key.
DELETE ADJACENT DUPLICATES FROM lt_zprft_ctr_unit_key.

* 3. Loop at all profit center and management unit combinations
LOOP AT lt_zprft_ctr_unit_key ASSIGNING <ls_zprft_ctr_unit_key>.

* 4. Select all relevant time intervals for the profit center currently processed
  LOOP AT lt_zprft_ctr ASSIGNING <ls_zprft_ctr>
                       WHERE /bic/zprft_ctr = <ls_zprft_ctr_unit_key>-zprft_ctr AND /bic/zmunit = <ls_zprft_ctr_unit_key>-zmunit.
                       "this is where the sorted-declaration of lt_zprft_ctr is of paramount importance for performance

    APPEND <ls_zprft_ctr> TO lt_zprft_ctr_tmp.
  ENDLOOP.


* 5. Select all relevant time intervals for the management unit currently processed
  LOOP AT lt_zmunit ASSIGNING <ls_zmunit> WHERE /bic/zmunit = <ls_zprft_ctr_unit_key>-zmunit.
                              " this is where the sorted-declaration of lt_zmunit is of paramount importance for performance
    APPEND <ls_zmunit> TO lt_zmunit_tmp.
  ENDLOOP.

* 6. PROVIDE
  PROVIDE
        FIELDS *
           FROM lt_zprft_ctr_tmp INTO ls_zprft_ctr_tmp
           VALID l_flg1
           BOUNDS datefrom AND dateto
        FIELDS *
           FROM lt_zmunit_tmp INTO ls_zmunit_tmp
           VALID l_flg2
           BOUNDS datefrom AND dateto
        BETWEEN '10000101' AND '99991231'.

* 7. Append relevant time-slices to result table
    IF l_flg1 IS NOT INITIAL.
        ls_ztarget-/bic/ztarget = ls_zprft_ctr_tmp-/bic/zprft_ctr.
        MOVE-CORRESPONDING ls_zprft_ctr_tmp TO ls_ztarget.
        MOVE-CORRESPONDING ls_zmunit_tmp TO ls_ztarget.
        APPEND ls_ztarget TO lt_ztarget.
        CLEAR ls_ztarget.
    ENDIF.
  ENDPROVIDE.

  REFRESH lt_zmunit_tmp.
  REFRESH lt_zprft_ctr_tmp.
  CLEAR ls_zmunit_tmp.
  CLEAR ls_zprft_ctr_tmp.
ENDLOOP. " end loop over all profit center / unit combinations

* Give out the initial tables and the resulting temporary join to the screen
WRITE: / 'Profit Center Table'.
SKIP.
WRITE: / 'ZPRFT_CTR_WORK', 20 'DATEFROM', 35 'DATETO', 50 'ZMUNIT'.
ULINE.
LOOP AT lt_zprft_ctr ASSIGNING <ls_zprft_ctr>.
  WRITE: / <ls_zprft_ctr>-/bic/zprft_ctr UNDER 'ZPRFT_CTR_WORK',
           <ls_zprft_ctr>-datefrom UNDER 'DATEFROM',
           <ls_zprft_ctr>-dateto UNDER 'DATETO',
           <ls_zprft_ctr>-/bic/zmunit UNDER 'ZMUNIT'.
ENDLOOP.

SKIP.

WRITE: / 'Management Unit Table'.
SKIP.
WRITE: / 'ZMUNIT_WORK', 20 'DATEFROM', 35 'DATETO', 50 'ZSEGMENT'.
ULINE.
LOOP AT lt_zmunit ASSIGNING <ls_zmunit>.
  WRITE: / <ls_zmunit>-/bic/zmunit UNDER 'ZMUNIT_WORK',
           <ls_zmunit>-datefrom UNDER 'DATEFROM',
           <ls_zmunit>-dateto UNDER 'DATETO',
           <ls_zmunit>-/bic/zsegment UNDER 'ZSEGMENT'.
ENDLOOP.


SORT lt_ztarget BY /bic/zprft_ctr datefrom dateto.
SKIP.
WRITE: / 'Target Table (ZTARGET)'.
SKIP.


WRITE: / 'ZTARGET', 20 'DATEFROM', 35 'DATETO', 50 'ZMUNIT', 70 'ZSEGMENT'.
ULINE.
LOOP AT lt_ztarget ASSIGNING <ls_ztarget>.
*WRITE <ls_ztarget>.
  WRITE: / <ls_ztarget>-/bic/ztarget UNDER 'ZTARGET',
           <ls_ztarget>-datefrom UNDER 'DATEFROM',
           <ls_ztarget>-dateto UNDER 'DATETO',
           <ls_ztarget>-/bic/zmunit UNDER 'ZMUNIT',
           <ls_ztarget>-/bic/zsegment UNDER 'ZSEGMENT'.
ENDLOOP.
